unittest
{
	assert( [ "3 7", "9 3 4" ].parse.expand.solve == "POSSIBLE" );
	assert( [ "3 5", "6 9 3" ].parse.expand.solve == "IMPOSSIBLE" );
	assert( [ "4 11", "11 3 7 15" ].parse.expand.solve == "POSSIBLE" );
	assert( [ "5 12", "10 2 8 6 4" ].parse.expand.solve == "IMPOSSIBLE" );
}



import std.algorithm;
import std.conv;
import std.numeric : gcd;
import std.range;
import std.stdio;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	auto nk = input.front.split.to!( long[] );
	input.popFront;
	auto as = input.front.split.to!( long[] );
	return tuple( nk[ 1 ], as );
}

auto solve( long k, long[] as )
{
	// 最大値と最大公約数を取得
	auto r = as.reduce!( max, gcd );
	if( k <= r[ 0 ] && k % r[ 1 ] == 0 ) return "POSSIBLE";
	else                                 return "IMPOSSIBLE";
}
